Calculus Single Variable
Brian E. Blank and Steven G. Krantz

Section 6.5
Applications of the Exponential Function

Load

Each execution group of this section must be entered before continuing with the worksheet.

```
> with(plots):  #Ignore any warnings!
Warning, the name changecoords has been redefined
```

1. A Basic Differential Equation

If A is any constant (usually positive in applications) and if λ is any positive constant, then the unique solution of the initial value problem

$$\frac{d}{dt}\mathrm{y}(t) = (-\lambda\,\mathrm{y}(t)) \quad , \quad \mathrm{y}(0) = A$$

is

$$\mathrm{y}(t) = A\,\mathrm{e}^{(-\lambda\,t)}.$$

The solution of this initial value problem is said to ***decay exponentially***.

2. Half-life

If a positive function f is decreasing, then we may call the function $t \to \tau(t)$ the halving time of f if

$$\mathbf{f}(t + \tau(t)) = \frac{\mathbf{f}(t)}{2}.$$

In general the halving time is nonconstant. However:

<u>Theorem</u> : Suppose that λ is a positive constant. The halving time τ of the exponential function $t \to A\,\mathbf{e}^{(-\lambda t)}$ is constant. More precisely, for every $0 < t$, the halving time is given by

$$\tau = \frac{1}{\lambda}\ \ln(2).$$

<u>Proof:</u>

```
> f := t -> A*exp(-lambda*t);
```

$$f := t \to A\,\mathbf{e}^{(-\lambda t)}$$

```
> tau = solve( f(t+tau) = f(t)/2, tau); #Note tau is independent
  of t
```

$$\tau = \frac{\ln(2)}{\lambda}$$

When $\mathrm{f}(t) = A\,\mathbf{e}^{(-\lambda t)}$ is the mass of an unstable isotope, then $\tau = \dfrac{\ln(2)}{\lambda}$ is said to be the ***half-life*** of the substance.

Theorem : **Suppose that λ is a positive constant. Then the exponential function $t \to A\, e^{(-\lambda t)}$ can be written as $t \to \frac{A}{2^{\left(\frac{t}{\tau}\right)}}$ where $\tau = \frac{1}{\lambda} \ln(2)$ is the half-life.**

Proof:

```
> formula1 := A*exp(-lambda*t);
```

$$formula1 := A\, \mathbf{e}^{(-\lambda t)}$$

```
> formula2 := A/2^(t/tau);
```

$$formula2 := \frac{A}{2^{\left(\frac{t}{\tau}\right)}}$$

```
> formula2 := subs(tau = ln(2)/lambda, formula2);
```

$$formula2 := \frac{A}{2^{\left(\frac{t\,\lambda}{\ln(2)}\right)}}$$

```
> simplify( formula1 - formula2 );
```

$$0$$

3. Exercises Involving Exponential Decay

Exercise

(Exercise 10, page 472) The mass of a radioactive sample is 10g in 1950 and 7 g in 2000. What is its half-life?

Solution

Let $t = 0$ correspond to 1950. Let $m(t)$ be the mass at time t.

```
> eqn1 := m(t) = 10*exp(-lambda*t);
```

$$eqn1 := m(t) = 10\, e^{(-\lambda\, t)}$$

```
> eqn2 := 7 = subs(t=50, rhs(eqn1));
```

$$eqn2 := 7 = 10\, e^{(-50\, \lambda)}$$

```
> eqn3 := lambda = solve(eqn2, lambda);
```

$$eqn3 := \lambda = -\frac{1}{50} \ln\left(\frac{7}{10}\right)$$

```
> eqn4 := tau = ln(2)/lambda;
```

$$eqn4 := \tau = \frac{\ln(2)}{\lambda}$$

```
> eqn5 := subs(eqn3, eqn4);
```

$$eqn5 := \tau = -\frac{50 \ln(2)}{\ln\left(\frac{7}{10}\right)}$$

```
> eqn6 := map(evalf, eqn5);
```

$$eqn6 := \tau = 97.16791050$$

4. Exercises Involving Radiocarbon Dating

Living matter contains two isotopes of carbon, ^{12}C and ^{14}C, in a known fixed ratio. After death, carbon is no longer metabolized and the amount of ^{14}C decreases due to radioactive decay. The mass of stable ^{12}C in a sample can be used to determine the mass m of ^{14}C that the sample had at the moment of death. Since the half-life of ^{14}C is known to be around 5700 years, the time of death can be calculated from the law of exponential decay. Willard Libby received the 1960 Nobel prize for physics for discovering this method of dating.

In each exercise we will let

$$\mathbf{y}(t) = \frac{m}{2^{\left(\frac{t}{5700}\right)}}$$

denote the amount of ^{14}C that exists t years after death.

Exercise
(Exercise 27, page 473)



The Shroud of Turin is a cloth believed by some to be the burial cloth of Jesus. In the 1980s, four strands of the cloth were given to four different institutions for dating. The results of the radiocarbon analysis were released on 13 October 1988. Each institution declared the Shroud of Turin to be a medieval forgery.

i) What percentage of m would the measured quantity of ^{14}C have been in order to authenticate the Shroud?
ii) The Shroud was dated to around 1325 C.E. Approximately what percentage of m was measured?

Solution:
The death date of Jesus is not precisely known. However, historical references indicate that it did not take place later than
36 C.E.. The fibers of a burial cloth could not have come from living matter at a later date.

i)

```
> s := 1988-36;   # time from death of Jesus to year of dating
```

$$s := 1952$$

```
> 100*y(s)/m*perCent = 100*evalf(1/(2^(s/5700)));
```

$$\frac{100\ \mathrm{y}(1952)\ perCent}{m} = 78.86974370$$

Any percentage substantially greater than 78.9 would have proved a forgery.

ii)

```
> s := 1988-1325;   # time from forgery to year of dating
```

$$s := 663$$

```
> 100*y(s)/m*perCent = 100*evalf(1/(2^(s/5700)));
```

$$\frac{100\ \mathrm{y}(663)\ perCent}{m} = 92.25405370$$

The measured amount was about 92.25 percent of m.

Exercise
(Exercise 28, page 473)


On 19 September 1991 a mummified human body was discovered in an Austrian glacier near the Tisenjoch pass.
Radiocarbon analysis has shown that the Iceman, as he has come to be called, died between 3350 B.C. and 3300 B.C.. What percentage of m was measured?

Solution:

```
> s := 1991+3325; # time from death of Iceman to radicarbon dating
```

$$s := 5316$$

```
> 100*y(s)/m*perCent = 100*evalf(1/(2^(s/5700)));
```

$$\frac{100\ \mathrm{y}(5316)\ perCent}{m} = 52.39018355$$

Exercise
(Exercise 29, page 473)

In 1994 previously unknown cave art was discovered near Chauvet, France. Radiocarbon dating established that the oldest images were drawn about 32410 years earlier. At the time of analysis, what percentage of m was the ${}^{14}C$ content of the pigments used in the oldest images?

Solution:

```
> s := 32410;
```

$$s := 32410$$

```
> 100*y(s)/m*perCent = 100*evalf(1/(2^(s/5700)));
  s := 's': #This restores T to its unassigned value after the problem
```

$$\frac{100\, \mathrm{y}(32410)\, perCent}{m} = 1.942468600$$

Exercise (Exercise 30, page 473)

The Chauvet Cave contains evidence that points to the earliest occurrence of art restoration. When samples, all having equal ^{12}C content, were extracted from pigments coming from different areas of a cave drawing, one sample had ^{14}C content that was 1.837 times greater than the ^{14}C content found in the other samples. About how many years after the original work was the ``touch-up" work done?

Solution:
The values of m are the same for both the newer and older pigments because they have the same amount of ^{12}C content. Suppose the older pigment was s years old. Then the ^{14}C content of the older sample was

$$\frac{m}{2^{\left(\frac{s}{5700}\right)}}.$$

Suppose that the touch-up work was done t years later. Then the ^{14}C content of the newer sample was

$$\frac{m}{2^{\left(\frac{s-t}{5700}\right)}}.$$

The information we are given may be expressed in the following equation:

```
> eqn :=   m/(2^((s-t)/5700)) = 1.837*m/(2^(s/5700));
```

$$eqn := \frac{m}{2^{\left(\frac{s}{5700}-\frac{t}{5700}\right)}} = \frac{1.837\, m}{2^{\left(\frac{s}{5700}\right)}}$$

We are to solve for *t*. There are other variables present, namely *m* and *T*, that we will not be able to determine. However, they will cancel.

```
> t = solve(eqn, t);
```

$$t = 5000.904271$$

Exercise (Exercise 31, page 473)

Cro-magnon skull, Neanderthal skull, Cro-magnon skull (28,000 years old)

Cro-magnon Man migrated to Europe about 40000 years ago. For a long time scientists believed that the Neanderthals vanished soon afterwards. However, recent radiocarbon dating of a Neanderthal fossil resulted in a ^{14}C value of 0.026m. For about how long did the European Neanderthals and Cro-Magnons coexist (at the least)?

Solution:

```
> fsolve(0.026 = 1/2^(t/5700), t);
```

$$30012.46403$$

Neanderthals and Cro-Magnons coexisted for at least 10,000 years.

Exercise (Exercise 32, page 473)

The original burial bundle, Smithsonian National Museum of Natural History

Spirit Cave Man, a partial mummy discovered in a Nevada cave in 1940, constitutes the oldest mummified remains that have been found in North American. Paleographic evidence led scholars to believe that the Spirit Cave Man lived about 2000 years ago. But in 1996 the Spirit Cave Man was scientifically dated for the first time. An analysis of hair samples revealed his age to be about 9400 years. In this instance, paleographic evidence had led to a 78.7% error. What percentage error in measuring the quantity of ^{14}C would have resulted in an underdating of 78.7 %?

Solution:A 9400 year old sample would have ^{14}C content equal to: $\frac{m}{2^{\left(\frac{9400}{5700}\right)}}$, or

```
> evalf(m/(2^(9400/5700)));
```

$$0.3188339565\, m$$

A 2000 year old sample would have ^{14}C content equal to: $\frac{m}{2^{\left(\frac{2000}{5700}\right)}}$, or

```
> evalf(m/(2^(2000/5700)));
```

$$0.7841071970\, m$$

If the former is correct then the percentage error of the latter is

```
> 100*(.7841071970*m-.3188339565*m)/(.3188339565*m)*percent;
```

$$145.9296386\ percent$$

Exercise (Exercise 33, page 473)

In 1941 the Dutch artist Han van Meegeren (1889-1947) painted *Christ and The Adultress*, which he sold as a Vermeer to German Reichsmarschall Hermann Göring. At that time, Meegeren's many Vermeer and de Hooch forgeries had already been accepted as genuine. After World War II, to avoid imprisonment for selling art work to the Nazi's, Meegeren confessed to having forged the paintings he sold. He was sentenced to prison for forgery but died prior to serving his sentence. Given that Vermeer died in 1675, what is the maximum percentage of m that can be found in the pigments of one of his paintings? Given that Meegeren turned to forgery in the 1930s, what is the minimum percentage of m that his fakes could have?

Solution:

In 2006 the percentage of m that a sample from a true Vermeer painting dating from 1675:

$\frac{100}{2^{\left(\frac{2006-1675}{5700}\right)}}$, or

```
> evalf(100.0/(2^((2006-1675)/5700)));
                                96.05481400
```

In 2006 the percentage of m that a sample from a Meegeren painting forged in 1930: $\frac{100}{2^{\left(\frac{2006-1930}{5700}\right)}}$

, or

```
> evalf(100/(2^((2006-1930)/5700)));
                                99.08006135
```

Copyright and Author Information

Worksheet Title: BlankKrantz-06_5b-R8.mws   A Maple Release 8 worksheet.

Author: Brian E. Blank
Date Created: 30 September 2006
Date Last Revised: 30 September 2007

For more information, please contact the author:

Department of Mathematics,
Washington University in St. Louis
St. Louis, MO   63130

Telephone: (314) 935-6763
e-mail: brian@math.wustl.edu